

DA RK



2D Baneberegning for passivt stabile raketter
Af
Hans Olaf Toft

Dansk Amatør Raket Klub, januar 1992, revideret januar 1998

Indhold:

1.0 INDLEDNING:	3
2.0 PROBLEMBESKRIVELSE:	4
2.1 BEVÆGELSEN I RAMPEN	4
2.2 DEN DREVNE FASE	5
2.3 DEN BALLISTISKE FASE	6
2.4 RECUPERATIONSFASEN	7
2.5 BEVÆGELSEN LANGT FRA RAMPEN	7
2.6 VINDENS INDFLYDELSE PÅ BANEN	8
2.7 DET FJERNEST MULIGE NEDSLAGSPUNKT	9
3.0 FYSISKE MODELLER:	12
3.1 RAKETMOTORENS FUNKTION	12
3.2 EN MODEL FOR TYNGDEACCELERATIONEN	14
3.3 LUFTMODSTANDEN	14
3.4 EN MODEL AF ATMOSFÆREN	16
3.5 FORHOLDENE I DE HØJERE LUFTLAG	19
3.6 EN MODEL FOR VINDEN	20
4.0 NUMERISK BANEBEREGNING:	22
4.1 EULER'S OG RUNGE-KUTTA'S METODER:	22
4.2 ET SIMPELT BANEBEREGNINGSPROGRAM	24
5.0 SYMBOLLISTE:	29
6.0 LITTERATURLISTE:	30

1.0 Indledning:

Denne note har til formål at sætte læseren i stand til at udføre en baneberegning for en passivt stabil raket, der bevæger sig gennem atmosfæren i op til 120 km's højde over jorden. Til dette formål gennemgås de forskellige faser i raketens færd, og der opstilles modeller for raketens fysiske omgivelser. Af praktiske grunde holdes beregningerne i 2 dimensioner. Begrebet passiv stabilitet dækker over, at raketten ikke indeholder mekanismer der aktivt kan regulere raketens kurs, men at kursen udelukkende er bestemt af affyringsrampens elevation og at raketens udformning sikrer at angrebsvinklen (vinklen mellem raketens hastighedsvektor og den længdekse) altid holdes nær 0 grader. De overordnede strukturer i et simpelt baneberegningssystem omtales, og nogle algoritmer for programmets procedurer gennemgås.

2.0 Problembeskrivelse:

En rakets flyvning kan, forudsat at raketten er stabil og ikke er underlagt en aktiv styring, inddeles i 4 faser, hvorfor et baneberegningsprogram må indeholde separate rutiner for hver af disse faser. De 4 faser er: bevægelsen i affyringsrampen, den drevne fase, den ballistiske fase, og sidst en eventuel recuperationsfase. De principielle bevægelsesligninger for de 4 faser skal nu opstilles, hvorefter de nødvendige fysiske modeller betragtes i kapitel 3. Bevægelsen antages at kunne beskrives tilstrækkeligt med en todimensional model, således at raketten bevæger sig i den plan der udspændes af affyringsrampens sigteretning og lodret. Den todimensionale beskrivelse vil normalt være tilstrækkelig god, når man primært interesserer sig for de maksimale værdier for højde og hastighed, hvorimod positionen for nedslagspunktet, som jo normalt kun har interesse ved ufredelige formål, ikke kan bestemmes videre nøjagtigt. Begrænsningen ved den todimensionale model er, at virkningen af Jordens rotation, ligesom vindpåvirkningerne ikke kan tages med i beregningerne.


2.1 Bevægelsen i rampen:

Så længe raketten befinder sig på rampen er den tvunget til at bevæge sig langs den rette linie med vinkel β , som rampen danner, med vandret. Banen søges beskrevet i et sædvanligt XY-koordinatsystem med origo lige under affyringsrampen, med Y-aksen lodret opad (antiparallelt med tyngreaccelerationen), og med X-aksen vandret. Fortegnet på X-koordinaten vælges positivt i rakettenes flyveretning. I dette koordinatsystem kan bevægelsesligningerne hurtigt opskrives ifølge figuren:

Raketten er underlagt tyngdekraften $[mg]$, der altid er rettet lodret nedad, reaktionskraften (thrusten) $[T]$ fra raketmotoren, der er parallel med raketten (og dermed rampen), og luftmodstanden (drag'et) $[D]$, der altid er rettet modsat bevægelsen, og derfor er antiparallel med rampen. Det antages, at raketten så sjældent er i reel fysisk kontakt med affyringsrampen, at man kan se bort fra yderligere friktion. Af Newton's 2. lov fås nu umiddelbart:

$$\dot{v} = \frac{T - D}{m} - g \cdot \sin(\beta)$$

$$\dot{s} = v$$



idet der er benyttet den sædvanlige notation¹ for acceleration, hastighed og sted². Disse differentiaalligningerne giver, sammen med begyndelsesbetingelserne $v(0) = 0$ og $s(0) = 0$, alle


¹Størrelser med prik angiver denne størrelses tidsafledede (differentialkoefficient).

²Som sædvanligt gælder: $a = \dot{v}$ $v = \dot{s}$ $a = \ddot{s}$


informationer om raketten bevægelse i rampen, idet de tidsafhængige størrelser T , D og m kan bestemmes jfr. kapitel 3. Det bemærkes, at s er den tilbagelagte vejstrækning, så bevægelsesligningerne kun er gyldige for $s < L$ (= rampens længde). Tyngdeaccelerationen g afhænger i princippet af raketten øjeblikkelige højde over Jorden, men kan antages at være konstant, så længe raketten befinder sig i affyrringsrampen.

2.2 Den drevne fase:

Når raketten forlader affyrringsrampen bliver bevægelsesligningerne mere komplicerede, fordi banen nu ikke længere tvinges til at følge en ret linie. Tværtimod er raketten nu et frit legeme, hvis øjeblikkelige sigteretning - attituden - $\Theta(t)$ ændrer sig. Samtidigt kan pludselige vindpåvirkninger slå raketten ud af kurs, så den øjeblikkelige hastighedsvektor danner vinklen $\alpha(t)$ - angrebsvinklen - med længdeaksen. Dette betyder, at man nu må udføre baneberegningen i (mindst) 2 dimensioner. Banen beskrives i det samme koordinatsystem som før. Alle relevante størrelser: de virkende kræfter, acceleration, hastighed og sted, beskrives med separate vandrette og lodrette komponenter. Thrust, og tyngdekraft er som før, men forholdene for luftmodstanden er nu mere komplicerede: Luftmodstanden må opdeles i en aksialkomponent F_A og en normalkomponent F_N . I det tilfælde, hvor normalkomponenten er 0, bliver aksialkomponenten lig med drag'et $[D]$. Normalkomponenten giver anledning til et drejningsmoment, og dermed en vinkelacceleration af størrelsen:



$$\ddot{\Theta} = \frac{F_N \cdot X_s}{I_R}$$



hvor I_R er raketten's inertimoment omkring en akse gennem tyngdepunktet, vinkelret på symmetriaksen, og X_s er afstanden mellem raketten's tyngdepunkt c_g og det resulterende angrebepunkt for luftens kræfter c_p . Beligheden af c_p og c_g vil variere med hastigheden og mængden af tilbageværende brændstof, og kan derfor ikke uden videre bestemmes. Mængden af tilbageværende brændstof påvirker også I_R . Endvidere vil F_N afhænge af både hastigheden og angrebsvinklen på kompliceret vis. Banen bestemmes desuden af differentialligningerne:

$$\dot{v}_x = \frac{(T - F_A) \cdot \cos(\mathbf{q}) - F_N \cdot \sin(\mathbf{q})}{m}$$

$$\dot{v}_y = \frac{(T - F_A) \cdot \sin(\mathbf{q}) + F_N \cdot \cos(\mathbf{q})}{m} - g$$

Vinkelaccelerationen bliver lader sig under normale omstændigheder ikke beregne, men må forsøges omgået. Størrelsen af drejningsmomentet afhænger af F_N som igen afhænger af angrebsvinklen α . Hvis raketten er stabil, søger drejningsmomentet at mindske angrebsvinklen. Det er derfor ikke urimeligt at udnytte, at stabiliteten vil holde α nær 0 under hele banen. Med antagelserne $\alpha = 0$ og $F_N = 0$ får vinkelaccelerationen ikke længere nogen betydning, og man behøver kun at tage hensyn til den aksiale luftmodstand. Det er muligt yderligere at simplificere bevægelsesligningerne, idet Θ kan elimineres ved brug af den resulterende fart v , og man får de endelige udtryk:

$$\dot{s}_x = v_x \quad \dot{s}_y = v_y$$

$$\cos(\mathbf{q}) = \frac{v_x}{v} \quad \sin(\mathbf{q}) = \frac{v_y}{v}$$

$$v = \sqrt{v_x^2 + v_y^2}$$

$$\dot{v}_x = \frac{T - D}{m} \cdot \frac{v_x}{v} \quad \dot{v}_y = \frac{T - D}{m} \cdot \frac{v_y}{v} - g$$

Den drevne fase begynder så snart raketten har forladt affyringsrampen ($t = t_r$), og begyndelsesbetingelserne for de ovenstående bevægelsesligninger bliver derfor:

$$v_x(t_r) = v(t_r) \cdot \cos(\mathbf{b}) \quad s_x(t_r) = L \cdot \cos(\mathbf{b})$$

$$v_y(t_r) = v(t_r) \cdot \sin(\mathbf{b}) \quad s_y(t_r) = L \cdot \sin(\mathbf{b})$$


2.3 Den ballistiske fase:

Indtil nu er kun beskrevet de principielle forhold der gælder mens raketmotoren brænder. Efter burnout følger raketten en ballistisk bane, det vil sige, at kun 'passive' kræfter (tyngdekraften og luftmodstanden) virker på raketten. Brændstoffet er nu brugt op, og motoren yder ikke længere nogen thrust, og bevægelsesligningerne må da være som i den drevne fase, blot med $T = 0$ og m konstant. Ligningerne for hastigheden bliver da:

$$\dot{v}_x = -\frac{D}{m} \cdot \frac{v_x}{v} \quad \dot{v}_y = -\frac{D}{m} \cdot \frac{v_y}{v} - g$$

mens ligningerne for positionen naturligtvis kan benyttes uændret som før. De ovenstående ligninger gælder i øvrigt uændret efter toppunktet, idet v_y skifter fortegn.

2.4 Recuperationsfasen:



Ved recuperering søges raketten nedslagshastighed mindsket så meget, at nedslag kan foregå under kontrollable forhold. En rimelig nedslagshastighed kan opnås ved brug af faldskærm, streamer, luftbremse eller blot ved en deling af raketten, så dens strømlinede form ødelægges. Fælles for alle disse fremgangsmåder er, at luftmodstanden søges øget. Bevægelsesligningerne bliver noget ændrede under denne fase, simpelthen fordi forholdene i x-retningen bliver noget uafklarede. En faldskærm vil yde nogen luftmodstand i både vandret og lodret retning. Faldskærmens virkning i lodret retning er ganske velkendt, mens dens egenskaber i vandret retning normalt regnes for ganske uinteressante. Under bevægelsen i faldskærm vil

hastigheden anpasse sig, så opdriften på faldskærmen bliver lige så stor som tyngdekraften på raketten. I princippet bliver den lodrette hastighed derfor konstant, mens den vandrette hastighed bliver 0, fordi der ikke virker andre kræfter end luftmodstanden i vandret plan. I praksis bestemmes den vandrette hastighed imidlertid af vinden, mens den lodrette hastighed vil variere med luftens densitet. Den lodrette bevægelse er bestemt af:

$$\dot{v}_y = \frac{E}{m} - g \quad E = \frac{1}{2} A_p c_{dp} v_y^2$$

$$\dot{s}_y = v_y$$


hvor [E] er opdriften på faldskærmen, A_p er faldskærmens plane areal og c_{dp} er faldskærmens drag koefficient. Som begyndelsesbetingelse ovenfor benyttes de værdier for s_y og v_y der gælder på det tidspunkt hvor faldskærmen separeres. Det tager ganske vidst principielt lidt tid før faldskærmen er helt udfoldet, men i praksis betyder det uhyre lidt for beregningerne.

2.5 Bevægelsen langt fra rampen:

Ved opstillingen af raketten bevægelsesligninger antages tyngdekraften at være antiparallel med Y-aksen. Det er derfor stiltigende forudsat, at Jordens radius er uendelig stor, da tyngdekraften altid vil være rettet mod Jordens centrum. Denne antagelse er kun rimelig så længe raketten holder sig 'i nærheden' af affyringsrampen. I denne anledning skal virkningen af Jordens krumning undersøges. Som det ses på figuren vil tyngdeaccelerationen nu have vinklen μ med Y-aksen, hvorfor tyngdeaccelerationen vil påvirke raketten i både X og Y retningen. Af hensyn til den sædvanlige fortegnskonvention for vinkler, skal μ have det modsatte fortegn af x (positiv omløbsretning). Korrektionen for Jordens krumning, forudsat at bevægelsen ikke overskrider den halve klode, opskrives ved en simpel geometrisk betragtning:

$$\mathbf{m} = -\text{Arctan}\left(\frac{x}{R_0 + y}\right)$$

$$\dot{v}_x = \frac{T - D}{m} \cdot \frac{v_x}{v} + g \cdot \sin(\mathbf{m}) \quad \dot{v}_y = \frac{T - D}{m} \cdot \frac{v_y}{v} - g \cdot \cos(\mathbf{m})$$



Bemærk, at fortegnet på x skyldes korrektionen skyldes fortegnskonventionen for μ . R_0 er Jordens radius på affyringsstedet. Da jordens overflade ikke længere følger X-aksen, må højden s_{y0} over Jorooverfladen beregnes separat:


$$s_{y0} = \sqrt{x^2 + (y + R_0)^2} - R_0$$

Nedslagspunktet er herved bestemt som $s_{y0} = 0$. Det skal understreges, at man virkelig skal langt væk fra rampen før Jordens krumning har

indflydelse. Fejlen der begås ved at antage en lodret tyngdeacceleration bliver størst ved nedslagspunktet, og man skal omkring 100 km væk fra rampen, for at dreje tyngdekraften blot en enkelt grad.

2.6 Vindens indflydelse på banen:

En egentlig model for vindens indflydelse på rakettenes øjeblikkelige position er meget vanskelig at opstille da vindhastigheden varierer tilfældigt. Under normale omstændigheder kan man imidlertid komme ret langt ved at antage at vindhastigheden er konstant i tid, men varierer som funktion af højden.



Under den drevne eller den ballistiske fase vil vinden kunne ændre rakettenes attitude. Tænkes vindens hastighed opdelt i en aksialkomponent, rettet direkte mod rakettenes hastighed og en normalkomponent vinkelret på rakettenes hastighed ses det at aksialkomponenten giver anledning til en (minimal) forøgelse af luftmodstanden, mens normalkomponenten vil give anledning til en kraft, som angriber raketten i det sædvanlige angrebspunkt for de aerodynamiske

kræfter c_p . Vinden giver altså anledning til en angrebsvinkel, som rakettenes egen aerodynamiske stabilitet vil eliminere ved at ændre attitude.

Antages rakettenes fart at være konstant, vil vindens påvirkning betyde, at rakettenes attitude ændrer sig. Rakettenes tyngdepunkt påvirkes imidlertid ikke af vinden, hvorfor banen ikke påvirkes.

Hvis raketten derimod accelererer, vil den ændrede attitude betyde, at accelerationsvektoren bliver drejet ind efter den nye attitude. Under den drevne fase betyder dette, at raketten vil dreje mod vinden. Normalt vælger man at lade raketten flyve imod vinden, da vindpåvirkningen så giver en fladere bane, og et eventuelt ballistisk nedslag længere væk fra affyringsrampen. Hvis raketten lander med faldskærm vil vinden istedet føre landingspunktet tilbage mod rampen. Den omvendte situation - hvor man skyder med vinden - vil give en stejlere bane end forventet, hvorfor raketten under nedstigning med faldskærm vil drive langt væk fra rampen, mens et eventuelt ballistisk nedslag vil forekomme på et tilfældigt sted i nærheden af rampen.

En kvantitativ beskrivelse af vindens indflydelse på banen er vanskelig at bestemme idet der kræves et indgående kendskab til raketens intertimoment og aerodynamiske normalcoeffcient. Imidlertid kan man let bestemme forholdene i de to grænsetilfælde, hvor raketens intertimoment er meget stort eller meget lille.

De fleste af DARK's raketter er ganske lange og ganske tunge, hvorfor deres intertimoment vil være stort. Vinkelaccelerationen bliver således ganske lille - i grænsetilfældet bliver den 0. Vinden påvirker dermed ikke banens forløb.

Raketter der har lille egenvægt vil ofte have ganske små intertimomenter, hvilket betyder at de kan udsættes for ganske store vinkelaccelerationer. Som følge heraf kan de antages at kompensere en eventuel angrebsvinkel momentant. Vinden vil således kunne få stor indflydelse på banen. Hvis vinden har hastigheden w (hvis fortegn er modsat den benyttede konvention, idet en stillestående raket vil bevæge sig med hastigheden $-w$ i forhold til luften), og w antages at være parallel med x -aksen, kan accelerationsvektoren fra afsnit 2.2 og 2.3 skrives som:

$$\dot{v}_x = \frac{T - D}{m} \cdot \frac{v_x - w}{\sqrt{(v_x - w)^2 + v_y^2}} \quad \dot{v}_y = \frac{T - D}{m} \cdot \frac{v_y}{\sqrt{(v_x - w)^2 + v_y^2}} - g$$

med $T = 0$ under den ballistiske fase jfr. afsnit 2.3. De øvrige bevægelsesligninger er uændrede.

2.7 Det fjernest mulige nedslagspunkt:

Af hensyn til sikkerheden er det vigtigt at en raketaffyring finder sted på et tilstrækkeligt stort afspærret testområde. Den nødvendige størrelse af testområdet afhænger af den aktuelle raket, vindforholdene og ikke mindst af sammentræf af begivenheder. Der skal derfor opstilles et estimat over hvor langt væk fra affyringsrampen en given raket kan lande. Det antages for simpeltheds skyld, at burnout finder sted i rampen $S_x = 0$ og $S_y = 0$ så raketten straks opnår burnothastigheden v_0 , der har retningen (elevationen) θ_0 med vandret. For simpeltheds skyld ses der også bort fra luftmodstanden, og raketens ballistiske position som funktion af tiden kan nu let opskrives:

$$s_x(t) = v_0 \cdot \cos(\theta_0) \cdot t$$

$$s_y(t) = v_0 \cdot \sin(\theta_0) \cdot t - \frac{1}{2} \cdot g t^2$$

$$s_x(t) = s_x(t^*) + w \cdot (t - t^*)$$

$$s_y(t) = s_y(t^*) - \frac{v_{eq}}{g} \cdot (t - t^*)$$

Til tidspunktet t^* frigøres en faldskærm, som får raketten til at falde med ligevægtshastigheden³ v_{eq} , mens den effektive sidevind w , der skal regnes som et gennemsnit over de relevante luftlag, vil få raketten til at drive i vandret retning med sidevindens hastighed. For $t > t^*$ gælder derfor:


Indføres sidevindskoefficienten $\zeta = w/v_{eq}$ fås det fjerneste nedslagspunkt, idet vinden tænkes at blæse i samme retning som den ballistiske bane:

$$S_{xmax} = s_x(t^*) + z \cdot s_y(t^*)$$

Det fjerneste nedslagspunkt ses at afhænge af raketten position til $t = t^*$, der kan bestemmes af ligningerne for den ballistiske position. Tidspunktet for faldskærmsfrigørelsen har afgørende indflydelse på nedslagspunktet, ligesom startelevationen θ_0 , hvorfor der må tages hensyn til de værste mulige værdier af disse størrelser.

Startelevation θ_0 må ikke forveksles med rampelevationen, fordi raketten kan skifte kurs på egen hånd på grund af et vindpust eller ved tab af en finne eller en anden form for forceret ustabilitet som er udenfor kontrol. Man kan derfor ikke bruge affyringsrampens elevation til under et uheldstilfælde at forudsige banens retning. Den startelevation θ_0 , som giver den længste bane viser sig at være givet ved:

$$q_0 = \text{Arctan}(z) \quad 0 \leq z < \infty \quad \& \quad 0^\circ \leq q_0 \leq 90^\circ$$



Med denne værste tænkelige startelevation indsat, og med udnyttelse af at der i det aktuelle vinkelinterval gælder:

$$\cos(\text{Arctan}(z)) = \frac{1}{\sqrt{1+z^2}} \quad \& \quad \sin(\text{Arctan}(z)) = \frac{z}{\sqrt{1+z^2}}$$

findes det værste tænkelige tidspunkt t^* for faldskærmens udfoldelse:

$$t^* = \frac{v_0}{z \cdot g} \cdot \sqrt{1+z^2}$$


som for $\zeta < 1$ bliver større end den ballistiske flyvetid. For det tilfælde at sidevinden er svag ($\zeta < 1$) er det ballistiske nedslagspunkt derfor det fjernest mulige, og det værste tænkelige tilfælde er den maksimale ballistiske rækkevidde $s_{xmax} = v_0^2/g$, som opnås ved $\theta_0 = 45^\circ$.

For $\zeta \geq 1$ findes det fjernest mulige nedslagspunkt:

³ Se evt. kapitel 3.3 for definition af ligevægtshastigheden

$$s_{xmax} = \frac{v_0^2}{2 \cdot g} \cdot \frac{1 + z^2}{z} \quad z \geq 1$$

Dette er raketten maksimale rækkevidde under sammenfald af de værst tænkelige omstændigheder. Det er ganske vidst meget usandsynligt at dette sammenfald forekommer i praksis, men testområdet skal principielt være stort nok til at indeholde denne situation, da et nedslag udenfor testområdet på alle måder vil være meget uheldigt. Tilfældet hvor faldskærmssystemet svigter er indeholdt i ovenstående, idet det skal regnes til det ballistiske tilfælde. Hvis vinden på affyringsdagen er svag benyttes den maksimale ballistiske rækkevidde uanset om faldskærmssystemet virker eller ej, og i kraftig vind regnes der for sikkerheds skyld med den relevante sidevinddrift. Det bemærkes at s_{xmax} altid bliver mindst det dobbelte af den højde som raketten ville opnå ved et lodret skud ($v_0^2/2g$).



Konklusionen på ovenstående bliver at testområdet skal være stort nok til at dække en cirkel hvis radius er værdien for s_{xmax} med affyringsrampen anbragt i cirkelns centrum. Der kræves således kendskab til vindforholdene på affyringsdagen, men til overslag kan man sætte $\zeta = 1$. I tilfælde af at raketten er udstyret med flere faldskærme, skal man regne med ligevægtshastigheden for den største (evt for samtlige) af faldskærmene.

3.0 Fysiske modeller:

I dette kapitel behandles rakettenes thrustfunktion [T], massefunktion [m] og luftmodstand [D]. Det bliver her nødvendigt at opstille modeller for de fysiske forhold i atmosfæren, idet denne udviser betydelige variationer som funktion af højden over jordoverfladen, årstiden og den geografiske position. Først beskrives funktionen af raketmotoren, siden behandles tyngdeaccelerationen, og sidst behandles luftmodstanden og forholdene i atmosfæren.

3.1 Raketmotorens funktion:

Raketmotoren kan opfattes som en katapult, der udslynger masse i en veldefineret retning. Den drivende mekanisme er næsten altid en kemisk reaktion (forbrænding), og den udslyngede masse er i praksis altid helt eller delvis på gasform. Tænkes et molekyle med massen m_0 at forlade raketmotoren med hastigheden U , har det pågældende molekyle fået tildelt impulsen:

$$I_0 = m_0 \cdot U$$

Da udstødningsmolekylet tidligere var i hvile (bortset fra kvantemekaniske effekter), må resten af raketten, som følge af princippet om impulsbevarelse, også tildeles en impuls af størrelse I_0 , men i modsat retning. Rakettens impulstilvækst er ensbetydende med en hastighedstilvækst, og dermed må raketten være blevet påvirket af en kraft, der kun kan stamme fra raketmotoren. Denne kraft kaldes sædvanligvis thrust (af udenlandsk oprindelse). Det kan med en vis rimelighed antages, at udstødningshastigheden U ikke varierer mens motoren brænder, og når alt brændstof, med massen m_p er opbrugt, har raketten totalt fået tildelt impulsen:

$$I_{tot} = \int_0^{t_{burn}} T(t) \cdot dt = m_p \cdot U$$

Rakettens løbende massetab pr. tidsenhed dm/dt kan bestemmes ved udnyttelse af ligningen for totalimpulsen og antagelsen om konstant udstrømningshastighed U . Et massetab dm giver nemlig en impulstilvækst dI :


$$dI = U \cdot dm = T(t) \cdot dt \Rightarrow$$

$$dm = m_p \cdot \frac{dI}{I_{tot}}$$

Som mål for en raketmotors brændstoføkonomi benytter man ofte den specifikke impuls I_{sp} :

$$I_{sp} = \frac{I_{tot}}{m_p \cdot g} = \frac{U}{g}$$

Traditionelt opgives thrusten i kilo, hvorfor specifik impuls måles i sekunder, og er oprindeligt defineret som den tid raketmotoren ville kunne yde en thrust på 1 kg med 1 kg brændstof til rådighed. Det ses, at udstrømningshastigheden ville være et (mindst) lige så godt mål for brændstoføkonomien, men benyttes traditionelt meget sjældent. Til brug for baneberegninger kan I_{sp} være et praktisk redskab til en eventuel skalering af thrusten.



For praktisk brug bestemmes thrust som funktion af tiden ved en statisk test af motoren. Under testen samples thrustfunktionen med tidsintervallet Δt , hvorefter man kender thrustens størrelse på samplingtiderne. Alle efterfølgende beregninger må således foretages i diskret tid med tidsafstanden Δt . Hvis samplingen foretages med passende omtanke er det principielt muligt senere at indføre et nyt samplingtidsinterval, som er Δt gange en rational faktor. I så fald bestemmes de nye værdier på thrustkurven ved digital gensampling. Skønt det er fuldt muligt at udføre digital gensampling i 'sand' tid under en baneberegning benytter man oftest

lineær interpolation eller splining, da det er lettere og fejlene ved disse useriøse fremgangsmåder alligevel bliver små. Den diskrete tid betyder, at kun tidspunkterne $k\Delta t$, hvor Δt er samplingtidsintervallet har mening, hvorfor man i reglen underforstår Δt og lader alle funktioner være afhængige af samplenummeret k .

Det relevante ligningssæt for beskrivelse af raketmotorens funktion bliver derfor efter diskretisering:

$$N = \frac{t_{burn}}{\Delta t} \quad \bar{T} = \sum_{k=0}^{N-1} T(k)$$

$$m(k+1) = m(k) - m_p \cdot \frac{T(k)}{T} \quad (0 < k < N)$$

endvidere kan det være praktisk at kunne beregne:

$$I_{tot} = \Delta t \cdot \sum_{k=0}^{N-1} T(k)$$

Ved visse typer af brændstof - navnlig zink/svovl - optræder der remanens. Remanensen er rester af brændstof, der af en eller anden grund er forblevet i motoren efter forbrændingen. For zink/svovls vedkommende skyldes fænomenet, at forbrændingen ikke frigører tilstrækkelig meget energi til at alt brændstof kan sublimere. Mængden af remanens er forskellig ved statiske og dynamiske test, og kan sættes til 10 hhv. 5% af brændstofmassen. Statische tests af lodret monterede motorer giver en remanens på 5%, svarende til ved dynamiske tests. Det må opfattes på den måde at ca. halvdelen af remanensen falder ud gennem dysen, når motoren er anbragt lodret. Konsekvensen for baneberegningen må være at den målte thrustkurve er rigtig, men at masseflowet er 5% mindre end angivet i formelen. Disse 5% af brændstofmassen må regnes payload.

For en raket, hvor motoren arbejder i en anden højde (et andet omgivelsetryk) end under den statiske test hvor thrustfunktionen er optaget, skal der desuden kompenseres for denne trykforskel:

$$T = T_{m\ddot{a}lt} + A_e (P_{ref} - P_{omgivelser})$$

hvor $T_{m\ddot{a}lt}$ er thrustfunktionen målt under en statisk test, P_{ref} er omgivelsetrykket under den statiske test, $P_{omgivelser}$ er raketens omgivelsetryk i den aktuelle højde og A_e er dysens exitareal. Det ses at

en given motors thrust vil stige med højden, og da motoren som sådan er uændret vil også den specifikke impuls stige med højden.

3.2 En model for tyngdeaccelerationen:

Da Jorden ikke er hverken rund eller homogen, vil tyngdeaccelerationen variere med den geografiske position. En model for tyngdeaccelerationens variation med breddegraden Φ er:

$$g = 9.780490 \cdot (1 + 0.0052884 \sin^2(\Phi) - 0.0000059 \sin^2(2\Phi))$$

Tyngdeaccelerationen aftager desuden med kvadratet på afstanden til Jordens centrum. Som værdi for Jordens radius R_0 kan man benytte $R_0 = 6371315$ m (middelradius) eller $R_0 = 6356912$ (polradius), hvis den lokale jordradius på affyringsområdet ikke kendes. Den øjeblikkelige værdi af tyngdeaccelerationen i højden s_y er da:

$$g(s_y) = g(0) \cdot \left(\frac{R_0}{R_0 + s_y} \right)^2$$

hvor $g(0)$ er den lokale værdi af tyngdeaccelerationen på jordoverfladen.

I praksis er R_0 dog så stor, at man op til 120 km med en fejl på mindre end 1% kan bruge:

$$g(s_y) = g(0) - 3.086 \cdot 10^{-6} \cdot s_y$$


3.3 Luftmodstanden:

Luftmodstanden beskrives normalt ved hjælp af Rayleigh's ligning:

$$D = \frac{1}{2} A \rho c_d \cdot v^2$$

hvor A er raketens tværsnitsareal vinkelret på flyveretningen, ρ er luftens densitet og c_d er raketens dragkoefficient.

Desværre er denne formel alt for simpel til at beskrive de reelle forhold. Problemerne skyldes, at strømningsforholdene, og dermed c_d , er stærkt hastighedsafhængige. En fornuftig fremgangsmåde må derfor være, at korrigere dragkoefficienten for den aktuelle hastighed. Den krævede funktionsbeskrivelse for dragkoefficienten afhænger helt af raketens form, og skal principielt bestemmes ved vindtunnelforsøg eller med avancerede aerodynamiske beregninger. Et praktisk alternativ går ud på i stedet at benytte andres målinger som en tilnærmet løsning. På denne måde fås en beskrivelse, der er kvalitativt rigtig, men ikke 100% nøjagtig.



Drag'et skyldes dels luftmolekylernes sammenstød med næsekeglen, dels overfladefriktion og dels det sug der opstår bag raketten fordi halesektionen ikke slutter 'pænt'. Imidlertid vil udstødningen fra raketmotoren fuldende raketten form, og dermed reducere luftmodstanden i den drevne fase. Sug'et vil da kun have indflydelse efter burnout, og mens motoren brænder, skal Λ korrigeres for *base-drag*. Figuren ovenfor viser derfor både Λ_{coast} for den ballistiske fase *coast-drag*, og Λ_{base} *base-drag*, som er den andel af drag'et, der skyldes ovennævnte sug. Disse data stammer fra sonderaketten *Tomahawk*, som i sin udformning minder meget om DARK's raketter.

Λ kan tilnærmelsesvis beregnes med følgende udtryk:

$$\Lambda_{\text{coast}}(M) = Q_2 \cdot M^2 + Q_1 \cdot M + Q_0$$

$$\Lambda_{\text{base}}(M) = W_2 \cdot M^2 + W_1 \cdot M + W_0$$

$$M = \frac{v}{c}$$

$$c_d = \begin{cases} c_{d0} \cdot (\Lambda_{\text{coast}} - \Lambda_{\text{base}}) & t < t_{\text{burn}} \\ c_{d0} \cdot \Lambda_{\text{coast}} & \text{ellers} \end{cases}$$

hvor c er lydens hastighed og M kaldes Mach-tallet. Q_0, Q_1 og Q_2 og W_0, W_1 og W_2 er passende konstanter, og c_{d0} typisk har værdier i området 0.3 - 0.6 og er rakettenes dragkoefficient ved lave hastigheder ($\approx 100\text{m/s}$).

De benyttede parametre er følgende:

	W_2	W_1	W_0
$0.000 \leq M \leq 0.500$	0	0	0.10438
$0.500 < M \leq 1.000$	0.90212	-0.92881	0.34242
$1.000 < M \leq 1.109$	-1.56757	3.11411	-1.23338
$1.109 < M \leq 1.250$	0.41145	-1.01499	0.91187
$1.250 < M \leq 2.375$	0.00094	0.00195	0.28210
$2.375 < M \leq 2.750$	-0.23649	1.1347	-1.06889
$2.750 < M \leq 5.300$	0.022547	-0.243841	0.7631

og

	Q ₂	Q ₁	Q ₀
0.000 ≤ M ≤ 0.375	0	0	1
0.375 < M ≤ 0.875	0.95952	-0.71263	1.13278
0.875 < M ≤ 1.197	1.31302	-1.56004	1.60401
1.197 < M ≤ 1.250	-24.619	59.25762	-34.03925
1.250 < M ≤ 2.250	0.29585	-1.40626	2.85864
2.250 < M ≤ 4.00	0.03793	-0.3643	1.81735
4.000 < M ≤ 10.0	0.007781	-0.143386	1.415353

Som det ses, kræves der kendskab til lydets hastighed c , der desværre afhænger af atmosfærens sammensætning og temperatur. For beregning af den egentlige luftmodstand kræves der endvidere kendskab til luftens densitet, der også afhænger af en række parametre. For videre beregninger kræves der derfor en model af atmosfæren.



Det ovenstående gælder generelt for legemer der bevæger sig gennem atmosfæren og derfor specielt også for det tilfælde hvor en raket recuperes med faldskærm. I denne situation antages faldskærmen på et givet tidspunkt at udfolde sig momentant, hvilket godt nok er urealistisk, men dog tilstrækkeligt beskrivende for situationen. Luftmodstanden på faldskærmen vil af naturlige årsager altid være langt større end den øvrige luftmodstand, som man nu med god samvittighed kan se bort fra. For korsfaldskærme har dragkoefficienten c_d værdien 0.85 - 0.9, hvis man benytter skærmens plane areal som referenceareal. Luftmodstanden på faldskærmen vokser med hastigheden, indtil den ligevægtssituation indtræder hvor tyngdekraften på raketten og opdriften på faldskærmen bliver lige store. Herefter er raketten hastighed den såkaldte ligevægtshastighed:



$$v_{eq} = \sqrt{\frac{2 \cdot m \cdot g}{r \cdot A_p \cdot c_{dp}}}$$

3.4 En model af atmosfæren:

På baggrund af en lang række målinger, er der fra forskellige civile og militære instansers side opstillet modeller for atmosfærens vigtigste parametre - ihvertfald hvad angår forholdene på den nordlige halvkugle. Modellerne er udgivet på både formel og tabelform, og er yderst velegnede til brug for baneberegninger.

De følgende grafer viser de vigtigste atmosfæreparametre på en skandinavisk sommerdag (60°N, juli måned):





Luftens densitet og lydets hastighed er de vigtigste parametre for baneberegning, og det bliver nødvendigt at indarbejde disse parametres variation i et generelt baneberegningssystem. En egentlig funktionsbeskrivelse af forholdene er ikke umiddelbart mulig, men de følgende tilnærmelser repræsenterer en praktisk og kompakt beskrivelse, som er god nok til alle almindelige formål.

Luftens densitet kan skrives som:

$$\rho = \exp(r_4 \cdot s_y^4 + r_3 \cdot s_y^3 + r_2 \cdot s_y^2 + r_1 \cdot s_y + r_0)$$

(s_y i meter)

hvor parametrene er højdeafhængige og givet ved:

Højde (km)	r_4	r_3	r_2	r_1	r_0
$0 \leq s_y < 25$	4.88158 E-18	-1.80805 E-13	2.43238 E-11	-9.69341 E-5	0.19223
$25 \leq s_y < 75$	-6.03361 E-19	1.03527 E-13	-5.74633 E-9	-2.21030 E-5	-0.39621
$75 \leq s_y < 120$	-1.00364 E-18	4.43973 E-13	-7.13702 E-8	4.77280 E-3	-1.2184 E2

Den viste approximation for densiteten viser sig at passe inden for et par procent i hele intervallet mellem 0 og 120 km's højde.

Lydets hastighed kan tilsvarende beskrives ved:

$$c = d_4 \cdot s_y^4 + d_3 \cdot s_y^3 + d_2 \cdot s_y^2 + d_1 \cdot s_y + d_0$$

(s_y i m)

Højde (km)	d_4	d_3	d_2	d_1	d_0
$0 \leq s_y < 10$	0	-2.91745 E-12	-6.35237 E-8	-3.05528 E-3	3.40234 E2
$10 \leq s_y < 23$	0	0	0	0	3.008 E2
$23 \leq s_y < 42$	-4.08891 E-16	5.26876 E-11	-2.46944 E-6	5.12093 E-2	-9.74348 E1
$42 \leq s_y < 82$	3.37071 E-17	-6.82406 E-12	4.10293 E-7	-6.72549 E-3	2.8723 E2
$82 \leq s_y \leq 120$	-1.12800 E-16	4.4066 E-11	-6.28567 E-6	3.90509 E-1	-8.69882 E2

som også viser sig at passe indenfor cirka 1 procent i hele intervallet.

Det skal understreges, at bedre approximationer kan findes (ved simpelt kurvefit), men så bliver antallet af beskrivende parametre også større.

3.5 Forholdene i de højere luftlag:

Ved større højder end 120 km bliver forholdene yderst vanskelige at beskrive, idet solens aktivitet, årstiden, Jordens magnetfelt m.m. vil påvirke forholdene. Luftens densitet viser sig at udvise voldsomme udsving, mens temperaturen, og dermed lydets hastighed (der er ganske vidst

efterhånden så langt mellem molekylerne, at 'lyd' er et tvivlsomt fysisk fænomen), er langt mere stabil. Det bemærkes, at luftens densitet i 120 km's højde er af størrelsesordenen 10^{-8} kg/m³, og drag'et på en raket med tværsnitsareal 1 m² og dragkoefficienten $c_d = 1$ vil være mindre end 1 Newton ved en næsten-orbital hastighed på 8 km/s. Dette drag vil derfor først være af betydning, hvis det får lov at virke gennem længere tid, hvilket vil være tilfældet for satellitter i lavt kredsløb. For satellitopsendelser, og specielt for sonderaketter, hvis hastighed vil være lille i disse højder, vil drag'et imidlertid være helt uden betydning, og der vil derfor ikke være nogen rimelighed i at tage hensyn til luftmodstanden i så store højder.

3.6 En model for vinden:

Vinden er af natur stokastisk, men da en egentlig stokastisk vindmodel bliver meget kompliceret, er det almindeligt at benytte middelhastigheden w som et mål for vinden. Almindeligvis løber vinden parallelt med jordoverfladen - lodret vind af betydning forekommer kun i forbindelse med skypumper, og disse lades i denne sammenhæng udenfor betragtning.


Indenfor et lokalt geografisk område, kan man regne med at vindretningen er uafhængig af højden over jordoverfladen, mens størrelsen af vindens hastighed varierer. Normalt vil vindens hastighed vokse med højden, indtil den når en konstant værdi i en vis højde. Som en første ordens betragtning kan vinden hen over et ujævnt terræn sammenlignes med forholdet i en grænselagsstrømning hen over en rug overflade. I en sådan strømning er hastigheden 0 ved selve overfladen, mens den har en konstant værdi i alle afstande større end grænselagstykkelsen. I selve grænselaget afhænger hastigheden af afstanden fra overfladen opløftet til en eller anden potens. Man kan dermed formulere den følgende model for størrelsen af vindhastigheden:

$$w = \begin{cases} k_w (y - y_0)^\lambda & 0 < y \leq y_g \\ w_g & y > y_g \end{cases}$$

hvor y_g er 'tykkelsen af grænselaget' og y_0 er højden af terrænnets grundniveau. Da affyringsrampen normalt ikke er nedgravet, vil y_0 i reglen være negativ. De to konstanter k_w og λ afhænger af terrænet.

I praksis bør man dog foretrække vindprofiler, som er fremkommet ved faktiske målinger. Vindprofiler måles ofte ved brug af vejrballon og teodolit.

Opstignings-hastigheden for en vejrballon kan antages at være rimeligt konstant, hvorfor en måling af sigtevinklerne til ballonnen i vandret og lodret plan er et mål for vindretning hhv. vindhastighed. Ofte benyttes en standardiseret måling, hvor en ballon af en standardiseret type fyldes med helium eller brint, til den netop kan bære en referencedødvægt. Når referencedødvægten fjernes, vil ballonnen holde en kendt opstigningshastighed, hvorfor teodolitens lodrette sigtevinkel direkte kan kalibreres i m/s. Figuren viser et eksempel på en vindprofil optaget ved brug af en vejrballon.



4.0 Numerisk baneberegning:

I dette kapitel omsættes baneberegningen til en numerisk opgave. Først gennemgås et par almindeligt forekommende metoder for numerisk løsning af differentiaalligninger, og siden følger et simpelt baneberegningssystem, der illustrerer det foregående. Programmet kan benyttes som det er eller bruges som idegrundlag for mere avancerede programmer. Den primære grund til at vedlægge et helt program er ikke at angive en ultimativ løsning, men derimod fordi det er den letteste måde at illustrere emnet på.

4.1 Euler's og Runge-Kutta's metoder:

Som eksempel betragtes differentiaalligningen:

$$\frac{dy}{dt} = e^y \cdot (3t^2 + 1)$$

hvor det forlanges at $y(0) = 0$. Standard fremgangsmåden består i at separere de variable, så al y - og t -afhængighed holdes på hver sin side af lighedstegnet:

$$e^y \cdot dy = (3t^2 + 1) \cdot dt$$

Der integreres nu på begge sider af lighedstegnet, under hensyntagen til begyndelsesbetingelsen:

$$\int_0^y e^y dy = \int_0^t (3t^2 + 1) dt + K$$

Begge disse integraler lader sig let beregne, og efter isolering af y fås:

$$y(t) = \ln(t^3 + t + 1)$$

som er den søgte løsning. Bemærk, at uden kendskab til begyndelsesbetingelsen ville der være uendeligt mange løsninger.

Den ovenstående differentiaalligning skal nu forsøges løst numerisk. En sådan numerisk løsning giver i heldigste fald nogle værdier på løsningsfunktionen, men aldrig en formel for løsningsfunktionen. Der findes et stort udvalg af numeriske fremgangsmåder. De bedste metoder er de, der giver en god tilnærmelse til løsningen med mindst muligt regnearbejde.

Den simpleste fremgangsmåde kaldes Euler's metode. For en differentiaalligning på formen:


$$\frac{dy}{dt} = f(t, y)$$

antages det at de infinitesimale størrelser dy og dt kan erstattes af nogle små men endelige størrelser Δt og Δy . For et givet valg af Δt og en givet startværdi kan løsningsfunktionen y bestemmes i

punkter med tidsmæssig afstand Δt . Når den k 'te løsningsværdi kendes kan den $k+1$ 'te løsningsværdi let beregnes:

$$y(k+1) = y(k) + \Delta t \cdot f(k, y(k))$$

hvor k , som tidligere nævnt, er kort skrivemåde for $k\Delta t$. Den fejl der begås ved beregningen af $y(k+1)$ vil desuden følge med ved beregningen af alle senere værdier. Den ophobede regnefejl vil således vokse efterhånden som punkterne på $y(t)$ bestemmes, og hvis resultatet skal være brugbart, er det nødvendigt at gå frem med meget korte skridt ($\Delta t =$ meget lille). Som demonstration af Eulers metode er løsningen til eksemplet bestemt ved et antal skridtlængder.




Hvor Eulers metode kun tager hensyn til hældningskoefficienten i det ene intervalendepunkt tager Runge-Kutta's metode hensyn til nogle passende valgte hældningskoefficienter inde i det betragtede interval. Herefter benyttes disse til beregning af en 'ækvivalent' hældningskoefficient som et vægtet gennemsnit af de interne hældningskoefficienter. Den færdige fremgangsmåde (for en 4. ordens metode) kan skrives på formen:

$$y(k+1) = y(k) + \frac{\Delta t}{6} \cdot (k_1 + 2k_2 + 2k_3 + k_4)$$

$$\begin{aligned} k_1 &= f(k, y(k)) \\ k_2 &= f(k + \frac{1}{2}, y(k) + \frac{1}{2} \cdot k_1 \cdot \Delta t) \\ k_3 &= f(k + \frac{1}{2}, y(k) + \frac{1}{2} \cdot k_2 \cdot \Delta t) \\ k_4 &= f(k + 1, y(k) + k_3 \cdot \Delta t) \end{aligned}$$

Det bemærkes i denne forbindelse, at $f(k+\frac{1}{2})$ principielt ikke har nogen mening, idet $(k+\frac{1}{2})\Delta t$ er et ikke eksisterende tidspunkt, midt mellem to samlinger. Denne mellemliggende værdi må bestemmes ved interpolation, enten lineær interpolation, splining eller digital gensampling.

Til sammenligning med Eulers metode er igen bestemt ved et antal skridtlængder. Runge-Kutta's metode kan normalt benyttes med noget større skridtlængder end Euler's. Til gengæld kræver den betydeligt længere regnetid, for den samme skridtlængde.



For begge metoder gælder at man med nogen fordel kan foretage en løbende ændring af skridtlængden. Dette kan enten gøres på basis af fine vurderinger af den maksimale fejl på løsningspunkterne, eller ved ganske simpelt at lade skridtlængden være proportional med den øjeblikkelige værdi af accelerationen. Variabel skridtlængde giver imidlertid problemer under den drevne fase, hvor thruskurven skal interpoleres.

4.2 Et simpelt baneberegningssystem:

De foregående kapitlers bevægelsesligninger og fysiske modeller benyttes nu i et simpelt baneberegningssystem. Programmet er fuldt funktionsdygtigt, og tager hensyn til de forskellige parametres afhængighed i forhold til omgivelserne. Af hensyn til programmets overskuelighed er der dog indført nogle begrænsninger. Det antages at opsendelsen finder sted i Danmark (56. breddegrad), at højden aldrig overstiger 10 km, og at hastigheden aldrig overstiger 0.85*lydens hastighed. Desuden ses der bort fra vindpåvirkninger og det antages at motoren yder en konstant thrust gennem hele sin brændtid. Imidlertid kan programmet let udvides til at behandle alle breddegrader, højder op til (mindst) 120 km og hastigheder op til 10*lydens hastighed. Programmet er skrevet i PASCAL, men indeholder ingen programmeringstekniske finurligheder og lader sig derfor let oversætte til andre sprog. Det skal understreges at programmet kun kan bestemme den ballistiske bane for en et-trins raket, men at en udbygning til flere trin, og recuperation med flere faldskærme kan indbygges.

```
program banesim;
```

```
(*****
*   Simpel 2D Baneberegning for 1 trins raketter           *
*   Hans Olaf Toft, december 1991                         *
*****)

var a,ax,ay      : real;   { acceleration   X Y & Total   }
    v,vx,vy      : real;   { hastighed      X Y & Total   }
    s,sx,sy      : real;   { afstand       X Y & Total   }

    mstart,mprop : real;   { startmasse og brændstofmasse }
    delta_t      : real;   { tidsstep           }
    tid,m,g,g0   : real;   { løbende tid, masse og tyngdeacc. }
    Lramp        : real;   { længde af rampe       }
    cd0          : real;   { dragkoefficient ved lav hastighed }
    drag         : real;   { løbende værdi af luftmodstand }
    areal        : real;   { tværsnitsareal for raket }
    tburn,ttimer : real;   { brændtid for raketmotor }
    beta        : real;   { rampeelevation }
    Thrust,Mflow : real;   { Reaktionskraft og masseflow for motor }
    Isp         : real;   { Specifik impuls }

(*****
* Function der bestemmer raketens dragkoefficient ud fra et *
* givet machtal.                                           *
* Funktionen gyldighedsområde er begrænset til M < 0.875. *
*****)
```



```

*****
function cd(M:real):real;

var base_corr : real;

function basedrag(x:real) : real;
var y : real;
begin
  if (x < 0.5) then y := 0.10438;
  if ((x >= 0.5) and (x < 0.875)) then
    y := 0.90212*x*x - 0.92881*x + 0.34242;
  basedrag := y;
end;

function coastdrag(x:real):real;
var z : real;
begin
  if (x < 0.375) then
    z := 1;
  if ((x >= 0.375) and (x < 0.875)) then
    z := 0.95952*x*x - 0.71263*x + 1.13278;
  coastdrag := z;
end;

begin
  base_corr := 0;
  if M < 0 then M := 0;
  if (tid < tburn) then base_corr := basedrag(M);
  cd := cd0 * (coastdrag(M) - base_corr);
end;

(*****
* Function der bestemmer Machtallet i en given højde [m] *
* Funktionen er begrænset til 0 - 10 km *
*****)

function Mach(x,v : real) : real;
begin
  if (x < 0 ) then x := 0;
  Mach := v/(-2.91745e -12*x*x*x - 6.35237e -8*x*x - 3.05528e -3*x +
3.40234e2);
end;

(*****
* Function der bestemmer luftens densitet i en given højde [m] *
* Funktionen er begrænset til 0 - 10 km *
*****)
function density(X:real) : real;
begin
  if (x < 0 ) then x := 0;
  density := exp(4.88158e -18*x*x*x*x - 1.80805e -13*x*x*x + 2.43238e -11*x*x
- 9.69341e -5*x + 0.19223);
end;

(*****
* Function der bestemmer tyngdeaccelerationen ved rampen *
*****)

function Gravity(x : real) : real;
var fi : real;
begin

```

```

    fi := pi*x/180;
    Gravity := 9.780490*(1 + 0.0052884*sin(fi)*sin(fi) - 0.0000059*
                sin(2*fi)*sin(2*fi))
end;

(*****
* Procedure der nulstiller baneparametre før start på beregning *
*****)
Procedure init;
var i : integer;
begin
    vx := 0;
    vy := 0;
    sx := 0;
    sy := 0;
    ax := 0;
    ay := 0;
    v := 0;
    s := 0;
    tid := 0;
end;

(*****
* Procedure til indlæsning af raketparametre *
*****)
procedure inddat;
var dia : real;

begin
    write('      STARTMASSE (kg)      ');
    readln(mstart);
    write('      BRÆNDSTOFMASSE (kg) ');
    readln(mprop);
    write('      DRAGKOEFFICIENT (*) ');
    readln(cd0);
    write('      RAKETDIAMETER (cm) ');
    readln(dia);
    write('      RAMPELÆNGDE (m)      ');
    readln(lramp);
    write('      ELEVATION (°)        ');
    readln(beta);
    write('      SPECIFIK IMPULS (s) ');
    readln(isp);
    write('      BRÆNDTID (s)         ');
    readln(tburn);
    write('      TIDSSTEP (s)         ');
    readln(delta_t);
    areal := pi*dia*dia*2.5e -5
end;

(*****
* Procedure for bevægelsen i rampen *
*****)
procedure ramp;
begin
    a := (Thrust - drag)/m - g*sin(beta);
    m := m - Mflow;
    v := v + a*delta_t;
    s := s + v*delta_t;
end;

```

```

(*****
* Procedure for den drevne fase
*****)
procedure boost;
begin
  ay := (Thrust -drag)*vy/(m*v) - g;
  ax := (Thrust -drag)*vx/(m*v);
  vx := vx + ax*delta_t;
  vy := vy + ay*delta_t;
  sx := sx + vx*delta_t;
  sy := sy + vy*delta_t;
  m := m - Mflow;
end;

(*****
* Procedure for den ballistiske fase
*****)
procedure coast;
begin
  ay := - drag*vy/(m*v) - g;
  ax := - drag*vz/(m*v);
  vx := vx + ax*delta_t;
  vy := vy + ay*delta_t;
  sx := sx + vx*delta_t;
  sy := sy + vy*delta_t;
end;

(*****
* Procedure der udfører baneberegning
*****)
procedure iterer;
var machtal,old_vy : real;
    old_time : real;
    Nburn : longint;
begin
  writeln('          Tid (s)      Højde (m)      Hastighed (m/s)',
          ' Acceleration (m/s^2) ');
  g0 := gravity(56);
  g := g0;
  m := mstart;
  Nburn := trunc(tburn/delta_t);
  tburn := Nburn*delta_t;
  Thrust := mprop*g*Isp/tburn;
  Mflow := delta_t*mprop/tburn;
  beta := pi/180*beta;

  writeln(' Start          ',tid:8:2,'          ',sy:10:2,'          ',
          ' sqrt(vx*vx+vy*vy):10:2,'          ',sqrt(ax*ax+ay*ay):10:2);

  old_time := 0;
  repeat
    ramp;
    sy := s*sin(beta);
    g := g0 - 3.086e-6*sy;
    tid := tid + delta_t;
    machtal := Mach(sy,v);
    drag := 0.5*density(sy)*v*v*areal*cd(machtal);
    if (( old_time < tburn) and (tid > tburn)) then
      writeln(' Burnout          ',tid:8:2,'          ',sy:10:2,'          ',

```

```

    sqrt(vx*vx+vy*vy):10:2,'          ',sqrt(ax*ax+ay*ay):10:2);
    old_time := tid;
until ((s >= Lramp) or (tid > tburn));
ax := a*cos(beta);
vx := v*cos(beta);
sx := s*cos(beta);
vy := v*sin(beta);
sy := s*sin(beta);
ay := a*sin(beta);
writeln(' Rampe          ',tid:8:2,'          ',sy:10:2,'          ',
        sqrt(vx*vx+vy*vy):10:2,'          ',sqrt(ax*ax+ay*ay):10:2);
repeat
    old_vy := vy;
    if (tid < tburn ) then boost;
    if (tid >= tburn) then coast;
    if ((old_vy >= 0) and ( vy <= 0)) then
        writeln(' Toppunkt          ',tid:8:2,'          ',sy:10:2,'          ',
                sqrt(vx*vx+vy*vy):10:2,'          ',sqrt(ax*ax+ay*ay):10:2);
        tid := tid + delta_t;
    if (( old_time < tburn) and (tid > tburn)) then
        writeln(' Burnout          ',tid:8:2,'          ',sy:10:2,'          ',
                sqrt(vx*vx+vy*vy):10:2,'          ',sqrt(ax*ax+ay*ay):10:2);
        old_time := tid;
    a := sqrt(ax*ax+ay*ay+0.0001);
    v := sqrt(vx*vx+vy*vy+0.0001);
    s := sqrt(sx*sx+sy*sy+0.0001);
    g := g0 - 3.086e-6*sy;
    machtal := Mach(sy,v);
    drag := 0.5*density(sy)*v*v*areal*cd(machtal);
until (sy < 0);
writeln(' Nedslag          ',tid:8:2,'          ',sy:10:2,'          ',
        sqrt(vx*vx+vy*vy):10:2,'          ',sqrt(ax*ax+ay*ay):10:2);
end;

(*****
* Hovedprogram
*****
begin
    init;
    inddat;
    iterer;
end.

```

5.0 Symbolliste:

A	: Tværsnitsareal.
A_e	: Tværsnitsarealet af dysen i dens exitplan.
a	: Acceleration.
α	: Raketten angrebsvinkel.
β	: Affyringsrampens elevation.
c	: Lydens hastighed.
c_d	: Luftmodstandskoefficient.
c_g	: Raketten massemidtpunkt.
c_p	: Resulterende angrebspunkt for luftens kræfter.
D	: Luftmodstand.
E	: Opdrift på faldskærm.
Δt	: Tidsstep (samplingtidsinterval).
F_N	: Luftmodstandens normalkomponent.
F_A	: Luftmodstandens aksialkomponent.
Φ	: Breddegrad.
g	: Tyngdeaccelerationen.
I	: Impuls.
I_{sp}	: Specifik impuls.
I_{tot}	: Totalimpuls.
k	: Tidsindex.
L	: Affyringsrampens længde.
M	: Mach-tallet.
m	: Raketten masse.
m_p	: Brændstofmasse.
R_0	: Jordens radius.
ρ	: Luftens densitet.
s	: Sted.
T	: Raketmotorens thrust.
t_{burn}	: Brændetid.
Θ	: Attitude (Banehældning).
v	: Hastighed.
v_x	: X-komponenten af v .
v_{eq}	: Ligevægtshastighed ved fald i faldskærm.
w	: Vindhastighed.
ζ	: Forholdet mellem vindhastighed og faldhastighed.

6.0 Litteraturliste:

Bulletin géodésique No 32, Paris, okt-dec. 1931.

C Steyaert.
Ballistic rocket flight.
Youth & Space 1981.

J Franck.
2-D baneberegningprogram for et-trins raketter.
Dansk Amatør Raket Klub.

H. D. Nissen
Missile stability and control. Part 1: Nonlinear 6-DOF rigid body model and reduced order control and stability model.
DTU 1996

D. B. Hansen
Raket Ståbi
En Håndbog for Raket-byggere
Dansk Amatør Raket klub 1993

Thiokol Chemical Corporation.
NASA Nike-Tomahawk handbook.
Contract NAS5-9693, Item No.6, Job order No. 721R67-04.

U.S. Standard Atmosphere Supplements, 1966.
Environmental Science Services Administration.
National Aeronautics And Space Administration.
United States Air Force.